
About this manual 
This is a shortened, hands-on manual that trains IFPRI and NARI analysts to build 
descriptive tables and cross-tabulations in Stata. It keeps every descriptive and data-
preparation step from Describing, Transforming, and Analyzing Data Using Stata: An 
Introductory Course.  

Who is it for. Analysts preparing summary tables from the producer survey. No prior Stata 
experience is assumed. 

How to use it. This is a workshop manual, not a lecture. Type each command in the 
Command window as you read. Questions are marked Practice, Question, or Challenge, 
and the expected answer follows in italics. 

Stata version. Written for Stata 18. Older versions may lack some commands; type help 
[command] or search online for the older equivalent. 

Practice datasets. The examples use the 2023 PNG Rural Household Survey files from the 
original course, so the printed answers can be checked: 

File Level (one row per…) Used in 
hh_3_4.dta (Module 3.4, housing quality) Household (2,699) Lessons 1–3, 5 
s11_long.dta (household roster) Person (16,192) Lessons 4–5 
lesson3.dta (compiled household file) Household (1,334) Lesson 4, outliers 

The same commands apply directly to the producer survey; only the file and variable 
names change. 

Lesson 1 – Getting to know Stata and your data 
Before building any table, know what is in the file: how many rows, what each variable 
means, and how it is stored. Open hh_3_4.dta by double-clicking it. 

1.1 The Stata screen 

The main screen has three windows: 



Window Name What it does 
1 Review and Results Shows the commands you ran and their output 
2 Command Where you type a command and press Enter 
3 Variables Lists every variable name and its label 

Figure 1.1: Main Stata screen 

 

1.2 Match variables to the questionnaire 

hh_3_4 corresponds to Module 3.4 (Housing quality) of the questionnaire. Question 
numbers and wording in the questionnaire match the variable names and labels in the 
Variables window. Keep the questionnaire open while you work: labels are cut off after 80 
characters, and the questionnaire shows the full question and its skip rules. 

Figure 1.2: Module 3.4 (Housing quality) from the 2023 PNG Rural Household Survey 



 

1.3 Browse the data 

The Data Browser shows the data as a spreadsheet. Open it any of three ways: 

1. Type browse in the Command window and press Enter. 
2. Menu: Data → Data Editor → Data Editor (Browse). 

3. Click the Data Browser icon in the toolbar:  

To view only some variables, list them: browse i9. 

Figure 1.3: Stata data browser screen 

 



1.4 Know how each variable is stored 

The storage type decides which table you can make from a variable. The Data Browser 
shows each type in a different color. 

Type 
Color in 
browser Example What it holds Typical table 

Numeric Black i4 – number of 
rooms 

Numbers with no 
labels 

summarize (mean, 
median) 

Categorical Blue i2 – floor 
material 

Numbers with a 
text label attached 
(6 = “Wood”) 

tabulate 
(frequencies, %) 

String Red i2_oth – “other” 
floor material, 
typed in 

Free text tabulate to read the 
responses, then 
recode 

Click a blue cell such as “Wood” and the number behind it (6) appears in the bar at the top 
of the browser. 

1.5 Count the observations 

count reports the number of observations (rows). 

Practice: How many observations are in this dataset? count → 2,699 households. 

Figure 1.4: Stata command and output for “count” 

 

1.6 Describe a variable with codebook 

codebook [varname] gives a one-variable overview: label, storage type, range, unique 
values, missing values, and example values. 

Practice: What do we know about drinking water? Type codebook i8a. 

Figure 1.5: Stata command and output for “codebook i8a” 



 

• What question was asked (the variable label)? 8a. What is the main source of 
drinking water for the people that live in your household? The label is cut off; check 
the questionnaire for the full wording. 

• What type of data is it? Numeric (int), with value labels, so it is categorical. 
• What is the range? 1 to 12. 
• How many observations are missing? 0. 
• How many unique values? 12. 
• What does one value mean? 6 = Unprotected spring. Only four of the 12 values are 

shown as examples; tab shows them all. 

Lesson 2 – One-variable descriptive tables 
Use tabulate for categorical variables (counts and percentages) and summarize for 
numeric variables (mean, spread, median). Dataset: hh_3_4.dta. 

2.1 Frequency tables with tabulate 

Code: tabulate [varname], shorthand tab [varname]. 

Many commands have a shorthand; this manual gives it the first time a command appears. 

Practice: How many households get drinking water from a protected dug well? tab i8a 

Figure 1.6: Stata command and output for “tab i8a” 



 

Column Meaning Example answer 
Freq. Number of households giving each 

answer 
184 households use a protected dug well 

Percent Share of all households giving each 
answer 

6.82% use a protected dug well 

Cum. Running total of the percentages, top 
to bottom 

97.70% use surface water or a source 
listed above it 

Question: How many households use surface water? 777. 

Question: How many households treat their water? tab i9 → 445 households (16.49%). 

2.2 Read the “other, specify” responses 

String variables hold the typed-in answers when a respondent chose “other”. Tabulate 
them to see what was written, which shows whether they should be recoded into existing 
categories. 

Question: What kind of variable is i2_oth? String. Check with codebook i2_oth, or see 
that it shows in red in browse. 

Practice: tab i2_oth. The list includes “BAMBU”, “Bamboo”, “Bambu”, “Bamboo floor” and 
“Bamboo flooring”: the same answer spelled several ways. These must be cleaned before 
they can go into a table. 

Figure 1.7: Stata command and output for “tab i2_oth” 



 

2.3 Visualize a distribution with histogram 

Code: histogram [varname], shorthand hist. Bar height shows how common each value 
is. 

Practice: hist i4 (number of rooms). 

Figure 1.8: Stata histogram output for “hist i4” 



 

Question: What is the most common number of rooms (the mode)? 2 rooms. Most houses 
have 1–5 rooms; very few have more than 5. 

tab i4 also works but gives one row per value. For a count variable like rooms, the mean 
and median are usually more useful. 

2.4 Summary statistics with summarize 

Code: summarize [varname], shorthand sum. Reports observations, mean, standard 
deviation, minimum and maximum. 

Practice: Average number of rooms? sum i4 

Figure 1.9: Stata command and output for “sum i4” 

 

• Average number of rooms? 2.87. 
• Challenge: Which earlier command also gave the minimum and maximum? 

codebook. 
• What is the median? This output does not show it. The maximum (10) is far above the 

mean, so a few large houses pull the mean up. The median is a better central value 
for skewed data. 

2.5 Add options: summarize, detail 

Most commands accept options, typed after a comma. The detail option (shorthand det) 
adds percentiles, the median, variance, skewness and kurtosis. 

Practice: sum i4, detail 



Figure 1.10: Stata command and output for “sum i4, detail” 

 

Question: What is the median number of rooms? 3 rooms (the 50% row). 

Lesson 3 – Cross-tabulations 
A cross-tabulation shows two categorical variables at once. The key skill is choosing the 
right percentage: cell, row or column each answer a different question. Dataset: 
hh_3_4.dta. 

3.1 Two-way tables 

Code: tabulate [varname1] [varname2]. The first variable forms the rows, the second the 
columns. 

Practice: Do households treat their water, by source? tab i8a i9 

Figure 1.11: Stata command and output for “tab i8a i9” 



 

Each cell counts the households with that combination of answers. 

Question: How many households collect rainwater and do not treat it? 549. 

3.2 Choose the right percentage 

Add an option after a comma to show percentages under each count. Read each one as “of 
[the group that sums to 100%], what share…”. 

Option 

Sums to 
100% 
across Answers the question 

tab i8a i9 example 
(rainwater, not treated) 

, cell The whole 
table 

What share of all households 
collect rainwater and do not treat 
it? 

20.34% (549 of 2,699) 

, row Each row Of households that collect 
rainwater, what share do not treat 
it? 

79.45% (549 of 691) 

, column 
(col) 

Each 
column 

Of households that do not treat 
their water, what share collect 
rainwater? 

24.36% (549 of 2,254) 

Practice: Run all three and find the rainwater/No cell in each. 

tab i8a i9, cell 
tab i8a i9, row 
tab i8a i9, col 



In each output, the top number in a cell is the count and the bottom number is the 
percentage. 

Figure 1.12: Stata command and output for “tab i8a i9, cell” (first rows) 

 

Figure 1.13: Stata command and output for “tab i8a i9, row” (first rows) 



 

Figure 1.14: Stata command and output for “tab i8a i9, col” (first rows) 



 

Rule of thumb for descriptive tables: put the grouping variable (region, sex, farm type) in 
the rows and use row percentages, so each row reads as a profile of that group. The order of 
the variables matters; tab i9 i8a, col gives the same numbers as tab i8a i9, row. 

3.3 Restrict a table with if 

if limits a command to observations that meet a condition. It uses these logical operators: 

Operator Meaning 
== equal to (two equals signs when comparing) 
~= or != not equal to 
> / >= greater than / greater than or equal to 
< / <= less than / less than or equal to 
& and 
\| or 
~ not 



Categorical variables are stored as numbers, so you need the number behind a label before 
writing the condition. Three ways to find it for “No” in i9: 

1. codebook i9 → shows that 2 = No. 
2. tab i9 then tab i9, nolabel → the nolabel option (shorthand nolab) prints the 

numbers instead of the labels. 
3. browse i9 → click a cell that says “No” and read the number in the top bar. 

Figure 1.15: Stata data browser window for “browse i9” 

 

Question: Using if, what share of households that do not treat their water use rainwater? 
tab i8a if i9==2 

Figure 1.16: Stata command and output for “tab i8a if i9==2” 



 

24.36%, the same as the column percentage above. The total is 2,254, not 2,699, because 
only households that do not treat their water are included. 

3.4 Practice: roof type by number of rooms 

Roof type is i1; number of rooms is i4. 

Question Command 
Roof type for households with more 
than 3 rooms 

tab i1 if i4>3 or tab i1 if i4>=4 

…with exactly 3 rooms tab i1 if i4==3 
…with 3 to 5 rooms tab i1 if i4>=3 & i4<=5 or tab i1 if 

i4>2 & i4<6 
…with fewer than 3 or more than 5 
rooms 

tab i1 if i4<3 \| i4>5 or tab i1 if 
i4<=2 \| i4>=6 

Challenge: Which households more often have a corrugated metal roof: those with more 
than 3 rooms, or those with 3 or fewer? 

tab i1 if i4>3 
tab i1 if i4<=3 

58.27% of households with more than 3 rooms have a corrugated metal roof, against 39.65% 
of households with 3 rooms or fewer. 

Caution: Stata treats a missing numeric value (.) as larger than any number. if x>3 
therefore includes missing values. Add & x<. (or & !missing(x)) when the variable has 
missing values; Lesson 4 shows why this matters. 



Lesson 4 – Preparing variables for tables 
Raw survey variables rarely go straight into a table. Missing-value codes distort means, 
unlabeled variables are unreadable, and many table rows need new indicator variables. 
Dataset: s11_long.dta (household roster), then lesson3.dta for outliers. 

4.1 Household-level and person-level data 
• Wide (household-level) data has one row per household, identified by hhid. 

hh_3_4 is wide. 
• Long data has several rows per household: one per person, per crop or per plot. 

You need two identifiers, such as hhid and a person ID (pid). The roster s11_long is 
long. 

Question: Will the roster have more or fewer observations than hh_3_4? More. count → 
16,192 people. 

Figure 2.1: Stata command and output for “count” in the roster 

 

Practice: How many men and women? Sex is a3, labeled 3. What is the sex of ${a1}?. 
The ${a1} part is the person’s name, filled in by the tablet and removed for anonymity. tab 
a3 → 8,459 male (52.24%) and 7,733 female (47.76%). tab is better than codebook here 
because it also gives percentages. 

Figure 2.2: Stata command and output for “tab a3” 

 

4.2 Find and fix missing-value codes 

Practice: Average age? Age is a4a. sum a4a 

Figure 2.3: Stata command and output for “sum a4a” 

 



A mean of 125 years is impossible. The maximum, 888, is the code enumerators entered 
when the respondent did not know the age. codebook a4a reports 0 missing, so Stata treats 
888 as a real age. 

Figure 2.4: Stata command and output for “codebook a4a” 

 

Stata marks missing values differently by storage type: 

Type Missing shown as Set to missing with 
Numeric . (a period) replace [varname]=. if … 
String "" (blank) replace [varname]="" if … 

a4a is numeric, so 888 becomes .. Use either command: 

recode a4a 888=. 
* or 
replace a4a=. if a4a==888 

Question: How many missing values now? codebook a4a → 1,904. Or use tab a4a, 
missing: the missing option adds a row for missing values in any tabulation. 

Figure 2.5: “codebook a4a” after recoding 888 to missing 

 

Question: Average age now? sum a4a → 23.43 years, from 14,288 observations. The 1,904 
people with unknown age are excluded. 

Figure 2.6: “sum a4a” after recoding 888 to missing 



 

Check the questionnaire or enumerator manual for every special code (888, 999, -99 and so 
on) before tabulating. 

4.3 Rename and label variables 

Tables print variable names and labels, so make them readable. 

Task Code Practice 
Rename a 
variable 

rename [old] [new] (ren) rename a4a age 

Label a 
variable 

label variable [varname] 
"label" (lab var) 

label variable age "Age of 
household member" 

The label must be in quotation marks. After renaming, use the new name (age) in every 
command. 

Practice: hist age. Most people in the sample are 25 or younger, and very few are over 80: a 
young population. 

Figure 2.7: Stata histogram output for “hist age” 

 

4.4 Create dummy (indicator) variables 

A dummy variable equals 1 if a condition is true and 0 if not. Its mean is the share of 
observations where the condition holds, which makes dummies the building blocks of most 
“% of households…” rows. 

Practice: Create child = 1 if the person is 15 or younger. 



Step Code (shorthand gen) 
Create the variable, all missing generate child=. 
Set to 1 for children replace child=1 if age<=15 
Set to 0 for adults replace child=0 if age>15 

Question: codebook child. How many missing values? 0, which is a problem. The 1,904 
people with unknown age were all set to 0 (adult), because . counts as larger than 15. 

Figure 2.8: Stata command and output for “codebook child” 

 

Fix: replace child=. if age==. Note the single = (assign a value) versus the double == 
(test a value). Better still, write the adult step as replace child=0 if age>15 & age<. so 
the error never happens. 

Figure 2.9: “codebook child” after assigning missing values 

 

4.5 Add value labels 

codebook child now shows 0, 1 and 1,904 missing, but no label says what 0 and 1 mean. 

1. Label the variable: label variable child "Is the household member 15 years 
or younger?" 

2. Define a set of value labels once: label define no_yes 0 "No" 1 "Yes". This is 
saved in memory and can be reused for any 0/1 variable. 

3. Attach it to the variable: label values child no_yes (shorthand lab val). 

Question: How many children are in the dataset? codebook child → 6,060 (8,228 adults, 
1,904 unknown). 



Figure 2.10: “codebook child” after adding labels 

 

Challenge: What share of children list “student” as their primary activity? tab a8 if 
child==1 → 86.39% (1,041 of 1,205 children with a reported activity). 

Figure 2.11: Stata command and output for “tab a8 if child==1” 

 

4.6 Bar graphs of prevalence by group 

Code: graph bar [varname], over([groupvar]). For a 0/1 variable, bar height is the 
share with value 1. 

Challenge: Which province has the highest share of children? 

graph bar child, over(prov) 
graph bar child, over(prov, label(angle(45))) 

Figure 2.12(a): “graph bar child, over(prov)” 



 

Figure 2.12(b): “graph bar child, over(prov, label(angle(45)))” 

 

The second line tilts the province names so they do not overlap. Morobe has the highest 
share of children aged 15 and under. sum child gives the same share as a number. 

Practice: Share of women by province: graph bar a3, over(prov). Every bar is above 1, 
so this is not a share. codebook a3 shows why: 1 = Male, 2 = Female. A 1/2 variable is not a 
dummy. 

Figure 2.13: “graph bar a3, over(prov)” 



 

Figure 2.14: Stata command and output for “codebook a3” 

 

Practice: Build a proper dummy: 

gen woman=. 
replace woman=1 if a3==2 
replace woman=0 if a3==1 
lab var woman "Is the household member a woman?" 
lab val woman no_yes 
codebook woman 
graph bar woman, over(prov) 

Figure 2.15: Stata command and output for “codebook woman” 



 

Figure 2.16: “graph bar woman, over(prov)” 

 

Milne Bay has the highest share of women in the sample. 

4.7 Check continuous variables for outliers 

Open lesson3.dta (1,334 households). Self-reported continuous variables such as land 
area often contain outliers from data-entry errors, misunderstood questions, or rough 
guesses. Outliers inflate means in a table. 

Practice: sum landholdings, det 

Figure 3.2: Stata command and output for “sum landholdings, det” 



 

1. Compare mean and median: 1.46 vs 1.05 hectares; the mean is about 40% higher, a 
sign of a right skew. 

2. Check minimum and maximum: 0 is plausible, but 19.55 hectares is far from the 
median. 

3. Visualize: hist landholdings shows a long right tail. 

Figure 3.3: Stata histogram output for “hist landholdings” 

 

How to treat outliers is a judgment call that depends on the question. In this exercise, 
values above the 99th percentile (8.0824) are replaced with the median (1.05): 

count if landholdings>8.0824          // 12 observations 
replace landholdings=1.05 if landholdings>8.0824   // (12 real changes made) 
hist landholdings 



Figure 3.4: Stata command and output for “count if landholdings>8.0824” 

 

Figure 3.5: Stata command and output for “replace landholdings=1.05 if 
landholdings>8.0824” 

 

Figure 3.6: “hist landholdings” after cleaning outliers 

 

The new histogram ends at 8 hectares instead of 20. Whatever rule your team uses, apply it 
the same way across tables and record it in the table notes. 

Lesson 5 – Changing the unit of analysis: collapse and merge 
A table describes one unit: households, people, plots or crops. collapse turns person- or 
plot-level data into household-level summaries; merge brings variables from different 
modules into one file so they can be cross-tabulated. Start from s11_long.dta with the age, 
child and woman variables built in Lesson 4. 

5.1 Collapse long data to the household level 

collapse replaces the data in memory with one row per group, holding the summary 
statistics you request: mean, median, sum, count, max, min. Save your person-level file first, 
because collapse overwrites it. 



Household-level indicator Statistic and variable 
Number of people in the household (count) pid 
Age of the youngest member (min) age 
Share of members who are children; share 
who are women 

(mean) child woman (the mean of a 0/1 
dummy is a share) 

Practice: 

collapse (count) pid (min) age (mean) child woman, by(hhid) 
count      // 2,699 households 

Figure 2.21: Variables window after collapsing pid, age, child and woman by 
household 

 

The variables keep their old names, and their labels become “(count) pid”, “(min) age” and 
so on. Rename and relabel them: 

rename pid hhsize 
lab var hhsize "Household size" 
rename age age_youngest 
lab var age_youngest "Age of the youngest household member" 
rename child perc_children 
lab var perc_children "Share of household members aged 15 or younger" 
rename woman perc_women 
lab var perc_women "Share of household members who are women" 

Question Command Answer 
Average household size sum hhsize 6.0 people (range 1–22) 
Average age of the youngest member sum age_youngest 8.7 years 
Average share of children per household sum perc_children 40.8% 
Average share of women per household sum perc_women 47.2% 

Figure 2.22: Stata command and output for “sum hhsize” 



 

(mean) child skips members with missing age, so perc_children is the share among 
members whose age is known. 

5.2 Merge two household-level files (1:1) 

Is household size related to roof type? Household size is in the collapsed roster; roof type 
(i1) is in hh_3_4. Both have one row per household, so each household matches exactly one 
row in the other file: a one-to-one merge. 

Figure 2.23: Example of a 1:1 merge, shown for one household (hhid) 

File hhid age_youngest i1 
s11_long (collapsed; n=2,699) 0100RSPN 11  
hh_3_4 (n=2,699) 0100RSPN  metal roof 
Merged data (n=2,699) 0100RSPN 11 metal roof 

Code: merge 1:1 [key variable] using "[file path]". See help merge for the full 
syntax. 

Figure 2.24: Stata help window for merge 

 

Practice: Merge on the household identifier hhid: 

merge 1:1 hhid using "C:\...\hh_3_4.dta" 

Figure 2.25: Stata command and output for “merge 1:1 hhid using …” 

 



All 2,699 households matched. Merge creates a variable _merge: 

_merge Meaning 
1 In the master file (the one already open) only 
2 In the using file (the one named in the command) only 
3 Matched in both 

Always check tab _merge before building tables, and investigate any 1s or 2s. Run drop 
_merge before the next merge, or Stata will refuse because _merge already exists. 

5.3 Summary statistics by group with bysort 

Challenge: Do households with thatched roofs or with corrugated metal roofs have a 
higher share of children? 

Option 1, one sum … if per group: 

sum perc_children if i1==1    // thatched: 43% 
sum perc_children if i1==2    // corrugated metal: 38% 

Option 2, one command for all groups. bysort [groupvar]: [command] repeats the 
command for each value of the group variable: 

bysort i1: sum perc_children 

Figure 2.26: Stata command and output for “bysort i1: sum perc_children” (abridged) 

 

Households with thatched roofs have a higher share of children (43%) than those with 
corrugated metal roofs (38%). bysort gives the same numbers with less typing, and works 
with tab, sum and most other commands. 

5.4 Merge household data onto person-level data (1:m) 

How many people live in a house with a thatched roof? This needs roof type on every 
person’s row. Start again: open hh_3_4.dta (household level) and merge the roster 
s11_long.dta (person level) onto it. One household matches many people, so the merge is 
one-to-many. 



Figure 2.27: Example of a 1:m merge, shown for one household (hhid) 

hh_3_4 (n=2,699) has one row for household 0100RSPN with i1 = metal roof. s11_long 
(n=16,192) has seven rows for it, one per person. The merged data (n=16,192) copies the 
roof type onto every person: 

hhid a_pid Sex i1 (added by merge) 
0100RSPN 1 M metal roof 
0100RSPN 2 F metal roof 
0100RSPN 3 F metal roof 
0100RSPN 4 F metal roof 
0100RSPN 5 M metal roof 
0100RSPN 6 M metal roof 
0100RSPN 7 M metal roof 
merge 1:m hhid using "C:\...\s11_long.dta" 
tab _merge      // all 16,192 observations: _merge==3 

Figure 2.28: Stata command and output for “merge 1:m hhid using …” 

 

If the person-level file is the one already open, the same merge is written merge m:1 hhid 
using "…\hh_3_4.dta". 

Question: How many people live in a house with a thatched roof? tab i1 → 8,328 people 
(51.43%). The data are now person-level, so tab counts people, not households. 

Challenge: How many women live in a house with a thatched roof? First confirm the 
coding (codebook a3: 2 = female), then use either approach: 

tab i1 if a3==2 
tab i1 a3, col 
tab a3 i1, row 

3,961 women (51.22% of women in the sample). 

The unit of the data decides the unit of the table. Before running any table, ask: “one row 
per what?” and label the table accordingly (“% of households” vs “% of people”). 

Getting help and quick reference 
Type help [command] to open the manual page for any command. Each page shows three 
things: 



• Syntax: how to write the command. For example, help tabulate twoway shows 
tabulate varname1 varname2 [if] [in] [weight] [, options]. Underlined 
letters are the shortest accepted abbreviation. 

• Options: what each option does, such as cell, row, column, missing, nolabel. 
• Examples: ready-to-adapt commands, such as tabulate region agecat, row. 

Figure 2.17: Stata help window for tabulate twoway – syntax 

 

Figure 2.18: Stata help window for tabulate twoway – options 

 

Figure 2.19: Stata help window for tabulate twoway – examples 

 

Figure 2.20: Stata help window for summarize – syntax and options 



 

Try help tabulate twoway, help summarize, help collapse and help merge. 

Command reference 
Task Command Shorthand Useful options 
Look at the data browse [varlist] br  
Count observations count [if]   
Describe one 
variable 

codebook [varname]   

Frequency table tabulate [var] tab missing, nolabel 
Cross-tabulation tabulate [var1] [var2] tab cell, row, column, 

missing 
Summary statistics summarize [varlist] sum detail 
Statistics by group bysort [groupvar]: 

[command] 
  

Histogram histogram [var] hist  
Bar graph of a 
share by group 

graph bar [dummy], 
over([groupvar]) 

 over(…, 
label(angle(45))) 

Restrict to a 
subgroup 

… if [condition]   

Recode a value recode [var] 
[old]=[new] 

  

Create a variable generate [new]=[value] gen  
Change values replace [var]=[value] 

if … 
  

Rename rename [old] [new] ren  
Variable label label variable [var] 

"text" 
lab var  

Define value labels label define [name] 0 
"No" 1 "Yes" 

lab def  



Task Command Shorthand Useful options 
Attach value labels label values [var] 

[name] 
lab val  

Summarize to one 
row per group 

collapse (stat) [vars], 
by([id]) 

 mean, median, sum, 
count, min, max 

Combine files merge 1:1 / 1:m / m:1 [id] 
using "[file]" 

 then tab _merge, drop 
_merge 

Delete a variable drop [var]   

Good practice for descriptive tables 
1. Know your data first: run count, codebook and tab before building anything. 
2. Recode special codes (888, 999) to missing before computing any mean. 
3. Use 0/1 dummies with value labels for every “% of…” row. 
4. Write & var<. whenever an if condition uses > or >=. 
5. Be explicit about the unit: households, people, plots or crops. 
6. Choose row, column or cell percentages to match the question the table answers. 
7. Check _merge after every merge. 


